

Problem 1.9-3 (1-19 in text): Fiber optic bundle

A fiber optic bundle (FOB) is shown in Figure P1.9-3 and used to transmit the light for a building application.




Figure P1.9-3: Fiber optic bundle used to transmit light.

The fiber optic bundle is composed of several, small diameter fibers that are each coated with a thin layer of polymer cladding and packed in approximately a hexagonal close-packed array. The porosity of the FOB is the ratio of the open area of the FOB face to its total area. The porosity of the FOB face is an important characteristic because any radiation that does not fall directly upon the fibers will not be transmitted and instead contributes to a thermal load on the FOB. The fibers are designed so that any radiation that strikes the face of a fiber is “trapped” by total internal reflection. However, radiation that strikes the interstitial areas between the fibers will instead be absorbed in the cladding very close to the FOB face. The volumetric generation of thermal energy associated with this radiation can be represented by:

$$\dot{g}''' = \frac{\phi \dot{q}''}{L_{ch}} \exp\left(-\frac{x}{L_{ch}}\right)$$

where $\dot{q}'' = 1 \times 10^5 \text{ W/m}^2$ is the energy flux incident on the face, $\phi = 0.05$ is the porosity of the FOB, x is the distance from the face, and $L_{ch} = 0.025 \text{ m}$ is the characteristic length for absorption of the energy. The outer radius of the FOB is $r_{out} = 2 \text{ cm}$. The face of the FOB as well as its outer surface are exposed to air at $T_\infty = 20^\circ\text{C}$ with heat transfer coefficient $\bar{h} = 5 \text{ W/m}^2\cdot\text{K}$. The FOB is a composite structure and therefore conduction through the FOB is a complicated problem involving conduction through several different media. Section 2.9 discusses methods for computing the effective thermal conductivity for a composite. The effective thermal conductivity of the FOB in the radial direction is $k_{eff,r} = 2.7 \text{ W/m}\cdot\text{K}$. In order to control the temperature of the FOB near the face where the volumetric generation of thermal energy is largest, it has been suggested that high conductivity filler material be inserted in the interstitial regions between the fibers. The result of the filler material is that the effective conductivity of the FOB in the axial direction varies with position according to:

$$k_{eff,x} = k_{eff,x,\infty} + \Delta k_{eff,x} \exp\left(-\frac{x}{L_k}\right)$$

where $k_{eff,x,\infty} = 2.0 \text{ W/m}\cdot\text{K}$ is the effective conductivity of the FOB in the x -direction without filler material, $\Delta k_{eff,x} = 28 \text{ W/m}\cdot\text{K}$ is the augmentation of the conductivity near the face, and $L_k = 0.05 \text{ m}$ is the characteristic length over which the effect of the filler material decays. The length of the FOB is effectively infinite.

a.) Is it appropriate to use a 1-D model of the FOB?

The inputs are entered in EES and functions are defined to return the volumetric generation and effective conductivity in the x -direction:

```
$UnitSystem SI MASS RAD PA K J
$TABSTOPS 0.2 0.4 0.6 0.8 3.5 in

function k_FOB(x)
  k_eff_x_infinity=2 [W/m-K]           "conductivity far from the face"
  L_k=0.05 [m]                         "characteristic length of elevated conductivity"
  Dk_eff_x=28 [W/m-K]                 "conductivity elevation at the face due to filler material"
  k_FOB=k_eff_x_infinity+Dk_eff_x*exp(-x/L_k) "conductivity"
end

function gv_FOB(x)
  phi=0.05 [-]                         "porosity"
  q``=1e5 [W/m^2]                     "incident heat flux"
  L_ch=0.025 [m]                      "characteristic length for absorption"
  gv_FOB=phi*q``*exp(-x/L_ch)/L_ch    "volumetric rate of thermal energy generation"
end

"Inputs"
k_eff_r=2.7 [W/m-K]                   "effective conductivity in the radial direction"
r_out=2 [cm]*convert(cm,m)            "radius of FOB"
h_bar=5 [W/m^2-K]                     "heat transfer coefficient"
T_infinity=converttemp(C,K,20[C])    "ambient temperature"
```

A Biot number is defined according to:

$$Bi = \frac{\bar{h} r_{out}}{k_{eff,r}} \quad (1)$$

which leads to $Bi = 0.037$, justifying an extended surface model of the FOB.

b.) Assume that your answer to (a) was yes. Develop a numerical model of the FOB.

Nodes are positioned along the FOB. The FOB is infinitely long; however, the first $L = 0.75$ m of the bundle is simulated. Examination of the solution shows that this is sufficient to capture the end effects.

```
L=0.75 [m]                           "length of FOB to simulate"
N=41 [-]                             "number of nodes"
Dx=L/(N-1)                           "distance between adjacent nodes"
duplicate i=1,N
  x[i]=Dx*(i-1)                       "position of each node"
end
```

An energy balance on node 1 leads to:

$$\bar{h} \pi r_{out}^2 (T_{\infty} - T_1) + \bar{h} 2 \pi r_{out} \frac{\Delta x}{2} (T_{\infty} - T_1) + \frac{\pi r_{out}^2}{\Delta x} k_{eff,x,x=(x_1+x_2)/2} (T_2 - T_1) + \dot{g}_{x=x_1} \pi r_{out}^2 \frac{\Delta x}{2} = 0 \quad (2)$$

```
h_bar*pi*r_out^2*(T_infinity-T[1])+h_bar*2*pi*r_out*(Dx/2)*(T_infinity-T[1])+&
pi*r_out^2*k_FOB((x[1]+x[2])/2)*(T[2]-T[1])/Dx+gv_FOB(x[1])*pi*r_out^2*Dx/2=0
"energy balance on node 1"
```

Energy balances on the internal nodes lead to:

$$\begin{aligned} \bar{h} 2 \pi r_{out} \Delta x (T_{\infty} - T_i) + \frac{\pi r_{out}^2}{\Delta x} k_{eff, x, x=(x_i+x_{i+1})/2} (T_{i+1} - T_i) + \\ \frac{\pi r_{out}^2}{\Delta x} k_{eff, x, x=(x_i+x_{i-1})/2} (T_{i-1} - T_i) + \dot{g}_{x=x_i}''' \pi r_{out}^2 \Delta x = 0 \end{aligned} \quad (3)$$

$$i = 2..(N-1)$$

```
duplicate i=2,(N-1)
h_bar*2*pi*r_out*Dx*(T_infinity-T[i])+pi*r_out^2*k_FOB((x[i]+x[i+1])/2)*(T[i+1]-T[i])/Dx+&
pi*r_out^2*k_FOB((x[i]+x[i-1])/2)*(T[i-1]-T[i])/Dx+gv_FOB(x[i])*pi*r_out^2*Dx=0
"energy balance on internal nodes"
end
```

The temperature of the last node is taken to be specified at the ambient temperature:

$$T_N = T_{\infty} \quad (4)$$

```
T[N]=T_infinity "node N temperature is specified"
```

The temperature is converted to Celsius:

```
duplicate i=1,N
T_C[i]=converttemp(K,C,T[i]) "temperature in C"
end
```

Figure 2 illustrates the temperature distribution within the FOB.




Figure 2: Temperature distribution within the FOB for the case where the filler material is present ($\Delta k_{eff,x} = 28 \text{ W/m-K}$) and the case where no filler material is present ($\Delta k_{eff,x} = 0$).

- c.) Overlay on a single plot the temperature distribution within the FOB for the case where the filler material is present ($\Delta k_{eff,x} = 28 \text{ W/m-K}$) and the case where no filler material is present ($\Delta k_{eff,x} = 0$).

Figure 2 shows the case where filler material is present and is removed. The reduction in the maximum temperature related to the addition of the filler material is evident in Figure 2.